

suppl file model (1).txt

\$PROBLEM PROJECT PAZOPANIB PK model
;UNITS: TAD=hour, Concentration=MG/L

\$INPUT ID DOSE TIME DV DAY TAD PKDAY MDV EVID AMT CMT SS II RATE ICLI IKENZ
IIC50 IVI CYP3 IGN
;RATE : infusion rate ifosfamide
;ICLI : Clearance ifosfamide
;IKENZ :rate constant for enzyme degradation
;IIC50 : ifosfamide concentration at 50% of maximum inhibition
;IVI : Volume of distribution ifosfamide
;CYP3 : CYP3A4*22 carriers =1 and non carriers = 0.

\$DATA pazopanib.csv
IGNORE=(IGN.EQ.1)

\$SUBROUTINE ADVAN6 TRANS1 TOL=4

\$MODEL
COMP=(DEPSN) ; 1 paz DOSE1
COMP=(CENTSN,DEFOBS) ; 2 paz Central
COMP=(PAZG2) ; 3 paz DOSE2
COMP=(IFO) ; 4 ifo central
COMP=(ENZ) ; 5 ifo enz
COMP=(PERIPH) ; 6 paz peripheral

\$PK
FL1=0
IF(CYP3.EQ.1) FL1=1
IF(CYP3.EQ.2) FL1=1

IF(PKDAY.EQ.1) IOV1=ETA(9)
IF(PKDAY.EQ.2) IOV1=ETA(10)
IF(PKDAY.EQ.3) IOV1=ETA(11)

CL = THETA(1)*EXP(ETA(1))* THETA(13)**FL1
V2 = THETA(2)*EXP(ETA(2))
KA1 = THETA(3)*EXP(ETA(3))
KA2 = THETA(4)*EXP(ETA(4))
ALAG3= THETA(5)*EXP(ETA(5))
Q24 = THETA(6)
V4 = THETA(7)*EXP(ETA(6))

KM = THETA(8)
VMAX = THETA(9)
LAMBDA=THETA(11)/24
DCRP = THETA(12)
TVB = (1-(DOSE-200)*VMAX/(KM+DOSE-200)) ; dose-dependent F
TVF1 = 1-DCRP+DCRP*EXP(-LAMBDA*TIME) ; time-dependent F
TVF = TVB*TVF1*EXP(ETA(7)+IOV1)

suppl file model (1).txt

TVFR = THETA(10)*EXP(ETA(8))
F1 = TVFR/(1+TVFR)*TVF ; fraction of fast abs
together with F
F3 = TVF-F1 ; fraction of
slow abs together with F

K20 = CL/V2
K26 = Q24/V2
K62 = Q24/V4
S2 = V2

;-----EBE of IFO-----

CLI = ICLI
KENZ = IKENZ
IC50I= IIC50
VI = IVI
K40 = CLI/VI
S4 = VI
A_0(5)=1

\$DES

CPI=A(4)/VI
DADT(1)=-KA1*A(1)
DADT(2)=KA1*A(1)+KA2*A(3)-K20*A(5)*A(2)-K26*A(2)+K62*A(6)
DADT(3)=-KA2*A(3)
DADT(6)=K26*A(2)-K62*A(6)

DADT(4)=-K40*A(5)*A(4)
DADT(5)=KENZ-KENZ*A(5)*(1-CPI/(CPI+IC50I))

\$ERROR

ENZ=A(5)
TY=F
IF(F.LT.0.001) TY=0.001
IPRED = TY
Y=IPRED*(1+ERR(1))+ERR(2)
IRES = DV-IPRED
IWRES = IRES/IPRED

\$THETA

(0,1) ; 1 CL
(0,2) ; 2 V2
(0,0.9) ; 3 KA1
(0,0.05); 4 KA2
(0,1.4) ; 5 ALAG3
(0,1) ; 6 Q24
(0,20) ; 7 V4
(0,200) ; 8 KMDOSE
(1 FIX) ; 9 VMAXDOSE

suppl file model (1).txt

(0,0.5) ; 10 TVFR
(0, 0.1); 11 LAMBDA
(0, 0.3); 12 DCRP
(0, 0.8); 13 CYP3

\$OMEGA

0.5 ; 1 CL
0 FIX ; 2 V2
0.5 ; 3 KA1
0 FIX ; 4 KA2
0 FIX ; 5 ALAG3
0.5 ; 6 V4
0.5 ; 7 F
0 FIX ; 8 TVFR

\$OMEGA BLOCK(1) 0.5 ; IOV F
\$OMEGA BLOCK(1) SAME
\$OMEGA BLOCK(1) SAME

\$SIGMA

0.05 ; prop paz
2 ; add paz

\$ESTIMATION METHOD=1 INTER NOABORT MAXEVAL=9999 SIG=3 PRINT=1 POSTHOC
MSFO=ms-f001
\$COVARIANCE PRINT=E MATRIX=S

\$TABLE ID TIME TAD DOSE DAY PKDAY CMT DV CWRES WRES IPRED PRED NOPRINT ONEHEADER
FILE=sdtab001